

```

10 ' FREEZER MANAGER PROGRAM
20 '
30 ' by
40 '
50 ' A. J. LAWLESS
60 '
70 CLEAR
80 SPEED WRITE 1
90 DIM item$(100),expirnum(100),mon(12)
100 GOSUB 600
110 MODE 1
120 PRINT "Type 'C' to create a new data file"
130 PRINT "or <any key> to load existing data"
140 a$=INKEY$:IF a$="" GOTO 140
150 IF UPPER$(a$)="C" GOTO 320
160 PRINT:PRINT
170 PRINT "Wind tape to start of data file"
180 OPENIN "FREEZERDAT": open and load data file
190 count=1
200 WHILE EOF=0
210 INPUT #9,item$(count),expirnum(count)
220 count=count+1
230 WEND
240 CLOSEIN
250 count=count-1
260 REM now print out contents of freezer
270 FOR n=1 TO count
280 PRINT item$(n);expirnum(n)
290 NEXT
300 FOR t=1 TO 2000:NEXT
310 REM Set up text windows
320 WINDOW #1,5,35,19,24:PAPER #1,2:PEN #1,3
330 WINDOW #2,1,40,1,1:PAPER #2,2:PEN #2,3
340 WINDOW #3,1,40,25,25:PAPER #3,2:PEN #3,3
350 WINDOW #4,1,40,2,24:PAPER #4,0:PEN #4,1
360 GOSUB 380
370 STOP
380 CLS
390 REM set up screen format and colours
400 PAPER 0:PEN 1:INK 0,10:INK 1,1:BORDER 1
410 DEFINT x,y : 'x and y made integers for screen positions
420 LOCATE 6,5:PRINT "FREEZER MANAGER - MASTER MENU"
430 x=7:y=8
440 LOCATE x,y:PRINT "Add Item to Freezer.....A"
450 LOCATE x,y+2:PRINT "Remove Item from Freezer.R"
460 LOCATE x,y+4:PRINT "Inventory.....I"
470 LOCATE x,y+6:PRINT "'Use by'.....U"
480 LOCATE x,y+8:PRINT "Save data to tape.....S"
490 INK 2,3:INK 3,6:'set up colours for window 1
500 CLS #1:LOCATE #1,5,3
510 PRINT #1,"Enter Option...";
520 ans$="ARIUS"
530 point=0:WHILE point=0:a$=INKEY$:REM set up pointer to correct routine
540 IF a$>" THEN point=INSTR(ans$,UPPER$(a$))
550 WEND
560 PRINT #1,a$:FOR t=1 TO 300:NEXT
570 ON point GOSUB 790,1020,1290,1590,1810
580 GOTO 380
590 RETURN
600 '*****
610 '
620 ' Routine to convert date into an integer number of days
630 ' elapsed since 1st Jan 1985
640 '*****
650 RESTORE 770
660 FOR n=1 TO 12
670 READ mon(n):REM Set up data array for days elapsed
680 NEXT
690 PRINT:PRINT
700 INPUT "Enter date (dd,mm,yy) : ",dd,mm,yy
710 IF dd<1 OR dd>31 THEN PRINT "Error in date !":GOTO 700
720 IF mm<1 OR mm>12 THEN PRINT "Error in month !":GOTO 700
730 IF yy<85 THEN PRINT "Error in year !":GOTO 700
740 PRINT:PRINT "DATE : ";dd;"-";mm;"-";yy
750 daynum=dd+mon(mm)+((yy-85)*365)
760 FOR t=1 TO 1000:NEXT
770 DATA 0,31,59,90,120,151,181,212,243,273,304,334
780 RETURN
790 '*****
800 '
810 ' Routine to ADD to contents of freezer
820 '
830 '*****
840 INK 1,0:REM Set up text colour for this option
850 CLS:PRINT
860 FOR t=1 TO 500:NEXT:REM delay to stop extraneous input
870 INPUT "Item to be added....";item$(count+1)
880 IF item$(count+1)="" GOTO 870
890 item$(count+1)=LOWER$(item$(count+1))
900 PRINT:INPUT "Max no. of weeks to be kept.. ",weeks%
910 IF weeks% <= 0 THEN RETURN
920 expirnum(count+1)=daynum+weeks%*7
930 count=count+1
940 PRINT:PRINT "Do you wish to add another item ?"
950 PRINT:PRINT:PRINT:PRINT "y = add another item, same date"
960 PRINT "c = add another item but change date"
970 PRINT:PRINT:PRINT " <Any other key will return to menu>"
980 a$=INKEY$:IF a$="" GOTO 980
990 IF UPPER$(a$)="C" THEN GOSUB 600
1000 IF UPPER$(a$)="Y" OR UPPER$(a$)="C" GOTO 850
1010 RETURN
1020 '*****
1030 '
1040 ' Routine to REMOVE item from freezer
1050 '
1060 '*****
1070 INK 1,0
1080 CLS
1090 FOR n=1 TO count
1100 PRINT n,item$(n)
1110 NEXT n
1120 PRINT:INPUT "Enter no. of item to be removed: ",no%
1130 IF no%<=0 OR no%>count THEN PRINT "ITEM DOES NOT EXIST !!" ELSE GOTO 1160
1140 FOR t=1 TO 1000:NEXT
1150 GOTO 1080
1160 PRINT:PRINT "Are you sure ? ";

```

```

1170 b$=INKEY$:IF b$="" GOTO 1170
1180 IF UPPER$(b$)<>"Y" THEN RETURN
1190 REM Squeeze arrays item$,expirnum to remove record
1200 WHILE noZ<=count
1210 item$(noZ)=item$(noZ+1)
1220 expirnum(noZ)=expirnum(noZ+1)
1230 noZ=noZ+1
1240 WEND
1250 count=count-1:REM Decrease inventory by one
1260 PRINT:PRINT "Done ..... remove anything else ? "
1270 a$=INKEY$:IF a$="" GOTO 1270
1280 IF UPPER$(a$)="Y" GOTO 1080 ELSE RETURN
1290 '*****
1300 '
1310 ' Routine to produce INVENTORY in alphabetical order.
1320 '
1330 '*****
1340 INK 0,0:INK 1,18:INK 2,9:INK 3,6
1350 CLS:CLS #2:CLS #3:CLS #4
1360 PRINT#2:PRINT #2,"ITEM","No. DAYS LEFT"
1370 ZONE 20:REM set up spacing for print columns
1380 n=1
1390 WHILE n<count
1400 IF item$(n) > item$(n+1) THEN GOSUB 1500
1410 n=n+1
1420 WEND
1430 FOR n=1 TO count
1440 PRINT #4,item$(n),expirnum(n)-daynum
1450 NEXT
1460 PRINT #3," Hit any key to Return to main Menu "
1470 IF INKEY$="" GOTO 1470
1480 ZONE 13:REM normal spacing reset
1490 RETURN
1500 REM***** SWAP subroutine *****
1510 temp$=item$(n)
1520 item$(n)=item$(n+1)
1530 item$(n+1)=temp$
1540 temp=expirnum(n)
1550 expirnum(n)=expirnum(n+1)

```

```

1560 expirnum(n+1)=temp
1570 n=0
1580 RETURN
1590 '*****
1600 '
1610 ' Routine to determine which items must be used up within
1620 ' a given period
1630 '
1640 '*****
1650 CLS
1660 PRINT "Any item listed below should be removed"
1670 nwks%=0
1680 FOR n=1 TO count
1690 IF expirnum(n) <= daynum+nwks%*7 THEN PRINT item$(n)
1700 NEXT
1710 PRINT:PRINT "Do you wish to see other expiry dates? ";
1720 a$=INKEY$:IF a$="" GOTO 1720
1730 PRINT a$
1740 FOR t=1 TO 1000:NEXT
1750 IF UPPER$(a$) <> "Y" THEN RETURN
1760 PRINT:INPUT "Enter no. of weeks to look ahead";nwks%
1770 PRINT:PRINT "The following items will be out of date "
1780 PRINT:PRINT " in ";nwks%;"weeks"
1790 FOR t=1 TO 1000:NEXT
1800 GOTO 1680
1810 '*****
1820 '
1830 ' Routine to save data to tape
1840 '
1850 '*****
1860 CLS
1870 SPEED WRITE 1
1880 OPENOUT "freezerdat"
1890 FOR n=1 TO count
1900 PRINT #9,item$(n):PRINT #9,expirnum(n)
1910 NEXT
1920 CLOSEOUT
1930 RETURN
1940 END

```